Problem:
Score : 400 points

Problem Statement

Given is a string S. Each character in S is either a digit (0, ..., 9) or ?.

Among the integers obtained by replacing each occurrence of ? with a digit, how many have a remainder of 5 when divided by 13? An integer may begin with 0.

Since the answer can be enormous, print the count modulo 10^9+7.

Constraints

S is a string consisting of digits (0, ..., 9) and ?.

1 \leq |S| \leq 10^5

Input

Input is given from Standard Input in the following format:

S

Output

Print the number of integers satisfying the condition, modulo 10^9+7.

Sample Input 1

??2??5

Sample Output 1

768

For example, 482305, 002865, and 972665 satisfy the condition.

Sample Input 2

?44

Sample Output 2

1

Only 044 satisfies the condition.

Sample Input 3

7?4

Sample Output 3

0

We may not be able to produce an integer satisfying the condition.

Sample Input 4

?6?42???8??2??06243????9??3???7258??5??7???????774????4?1??17???9?5?70???76???

Sample Output 4

153716888

Solution in Fortran (gfortran v4.8.4) (judge verdict: Accepted)

program digits_parade
  implicit none
  integer(8), parameter :: md = 1000000007_8
  character(100000) :: s
  integer :: n, i, j, k, l = 0
  integer(8) :: dp(0:1,0:12) = 0_8
  read(*,*) s
  s = trim(adjustl(s))
  n = len_trim(s)
  dp(0,0) = 1_8
  do i = 1, n
    l = 1-l
    dp(l,:) = 0_8
    if (s(i:i) == "?") then
      do k = 0, 9
        do j = 0, 12
          dp(l,mod(10*j+k,13)) = mod(dp(l,mod(10*j+k,13))+dp(1-l,j),md)
        end do
      end do
    else
      k = ichar(s(i:i))-48
      do j = 0, 12
        dp(l,mod(10*j+k,13)) = dp(1-l,j)
      end do
    end if
  end do
  write(*,'(i0)') dp(l,5)
  stop
end program digits_parade